
INTRODUCTION

# ONE OF FIRST STEPS IN DATA ANALYSIS ... A DECISION
ON HOW TO HANDLE MISSING VALUES

DELETE OBSERVATIONS WITH EXCESS MISSING DATA 

DELETE VARIABLES WITH EXCESS MISSING DATA

SUBSTITUTE IMPUTED VALUES FOR MISSING VALUES

TAKE NO ACTION AT ALL IF THE AMOUNT OF MISSING DATA IS
INSIGNIFICANT AND NOT LIKELY TO AFFECT THE ANALYSIS 

# DECISION FACILITATED BY KNOWING THE AMOUNT
OF MISSING DATA

# USE AN ODS OUTPUT STATEMENT, PROC FREQ, AND
SOME DATA STEP PROGRAMMING 

PRODUCE A MISSING DATA REPORT SHOWING THE
PERCENTAGE  OF MISSING DATA FOR EACH VARIABLE IN A
DATA SET  

IDENTIFY AND DROP ALL VARIABLES/OBSERVATIONS WITH
EITHER ALL OR A HIGH PERCENTAGE OF MISSING VALUES

MAKE SOME DATA FOR EXAMPLES

# USE SASHELP.CLASS ...

* EXAMPLE 1;
data class;
set sashelp.class;
if ranuni(987) le .5;
if ranuni(987) le .2 then
   call missing(weight);
if ranuni(987) le .1 then 
   call missing(sex, age);
run;

REVIEW

# PROC FREQ DEFAULT

proc freq data=class;
run;

# TABLES OF ALL VARIABLES

# COUNTS (NO PERCENTAGES) OF MISSING DATA

# ADD A TABLES STATEMENT AND THE MISSING
OPTION ...

proc freq data=class;
tables _all_ / missing;
run;

# COUNTS AND PERCENTAGES OF MISSING DATA

# BETTER IF NON‐MISSING DATA ALL IN ONE GROUP

# USE A FORMAT TO CREATE GROUPS OF MISSING AND
NON‐MISSING DATA

proc format;
value nm low‐high = 'OK' other = 'MISSING';
value $ch ' ' = 'MISSING' other = 'OK';
run;

# LOW‐HIGH in NM FORMAT AVOIDS HAVING TO SPECIFY THE RANGE
OF ALL POSSIBLE MISSING NUMERIC VALUES ... REMEMBER THERE
ARE 27 OF THEM (._ , . , .A through .Z)

proc freq data=class;
tables _all_ / missing;
format _numeric_ nm.  _character_ $ch.;
run;

# OK FOR A DATA SET WITH ONLY  FEW VARIABLES

# WRITE RESULTS TO A DATA SET ...
FORMATTED

proc freq data=class;
tables _all_ / noprint missing out=tables;
format _numeric_ nm.   _character_ $ch.;
run;

UNFORMATTED
# DOES NOT WORK ... ONLY THE

LAST VARIABLE IS IN THE OUTPUT
DATA SET (WEIGHT)

ALTERNATIVE ... ODS OUTPUT

CREATE DATA SET ... TABLES

proc format;
value nm low‐high = '1' other = '0';
value $ch ' ' = '0' other = '1';
run;

ods listing close;
ods output onewayfreqs=tables;

proc freq data=class;
tables _all_ / missing;
format _numeric_ nm.  _character_ $ch.;
run;

ods output close;
ods listing;

# TABLE OF ALL VARIABLES

# "F_" VARIABLES HAVE FORMATTED VALUES (0, 1)

# CHANGE ODS OUTPUT STATEMENT ...

ods output onewayfreqs=tables (keep=table frequency percent f_:);

ADD A DATA STEP

data report;
length var $32; 
do until (last.table); 

set tables; 
by table notsorted;

  array names(*) f_:  ;
  select (names(_n_)); 
   when ('0') do; miss = frequency; p_miss = percent; end;
  when ('1') do; ok = frequency;  p_ok = percent; end;

end;
end;

miss = coalesce(miss,0)               /* change  missing    values   to zero */
pmiss = coalesce(pmiss, 0);
ok = coalesce(ok, 0);
p_ok = coalesce(p_ok, 0);
var  = scan(table,‐1); 

format miss ok comma7. p_: 5.1 ;
label miss  = 'N_MISSING'

ok  = 'N_OK'
p_miss = '%_MISSING'
p_ok  = '%_OK'
var = 'VARIABLE';

keep var miss ok p_:  ; 
run;

# A MISSING DATA
REPORT ...

# TRY SAME SAS CODE WITH ANOTHER DATA SET ...
SASHELP.HEART

# THE SAS CODE IS
"GENERIC"

# THE ONLY CHANGE IS
THE NAME OF THE
DATA SET

ADD VARIABLE LABELS

# CHANGE THE LENGTH STATEMENT AT THE TOP OF
THE DATA STEP ...

length var $32 label $100; 

# ADD A STATEMENT ...

var  = scan(table,‐1);
* new statement;
label  = vlabelx(var);
keep var label miss ok p_:  ;

MISSING DATA PATTERNS

# USE PROC MI ... NUMERIC VARIABLES

proc mi data=class;
ods select misspattern;
run;

# USE PROC FREQ WITH A LIST OPTION ... ALL VARS

proc format;
value nm low‐high = 'X' 
                 other = '.';
 value $ch ' ' = '.' 
                   other = 'X';
run;

proc freq data=class;
table Name*Sex*Age*Height*Weight / list missing nocum;
format _numeric_ nm. _character_ $ch.;
run;

# USE PROC SQL IF THERE ARE LOTS OF VARIABLES ...

proc sql noprint;
select name into :vars separated by '*'
from dictionary.columns
where libname eq 'WORK' and memname eq 'CLASS' ;
quit;

proc freq data=class;
tables &vars / list missing nocum out=miss;
format _numeric_ nm. _character_ $ch.;
run;

# MACRO VARIABLE ... Name*Sex*Age*Height*Weight

# DROP VARIABLES WITH EXCESS MISSING DATA

# USE VALUES FROM THE MISSING DATA REPORT ...

proc sql noprint;
select var into :droplist separated by ' '
from report 
where p_miss ge 50;
quit;

data heart;
set sashelp.heart(drop=&droplist);
run;

(data set SASHELP.HEART ... DeathCause, AgeCHDDiag, AgeAtDeath)  

# DROP OBSERVATIONS WITH TOO MANY MISSING
VALUES ...

data heart;
set sashelp.heart;
if cmiss (of _all_) ge 5 then delete;
run;

# CMISS FUNCTION ALLOWS BOTH CHARACTER AND
NUMERIC VARIABLES (NMISS ... ONLY NUMERIC)

# DROP OBSERVATIONS WITH ALL MISSING VALUES ...

options missing=' '; /* CATS with numeric data */

data new;
set old;
if missing(cats(of _all_)) then delete;
run;

options missing='.';

NLEVELS

# PROC FREQ WITH NLEVELS OPTION ... A QUICK WAY
TO DROP VARIABLES WITH ALL MISSING VALUES

# SOME DATA ...

data test;
input name :  $10.  gender  :  $1.  age  height weight;
datalines;
MIKE . 21 60 .L
MARY . .Z 54 .H
MARK . 45 10 .H
;

# NUMBER OF
UNIQUE VALUES
FOR EACH
VARIABLE

ods output nlevels=tables;

proc freq data=test nlevels;
run;

ods output close;

# WHAT IF NO MISSING DATA

data test;
input name : $10. gender : $1. age height weight;
datalines;
MIKE M 21 72 200
MARY F 56 54 120
MARK M 45 60 110
;

proc format; 
value nm  ._ ‐ .z  = . other = 'OK' ;
value $ch '  '  = '  ' other = 'OK';
run;

ods listing close; 
ods output nlevels=tables;

proc freq data=test nlevels; 
ods select nlevels;
format _numeric_ nm.   _character_ $ch.;
run;

ods output close; 
ods listing;

* error in SQL if no missing data;   

proc sql noprint; 
select tablevar into :droplist separated by ' '
from tables
where nnonmisslevels eq 0;
quit;

data new;
set test (drop=&droplist); 
run; 

REPLACE NUMERIC MISSING WITH 0

data test;
input name :  $10.  gender  :  $1.  age  height weight;
datalines;
MIKE . 21 60 .L
MARY . .Z 54 .H
MARK . 45 10 .H
;

data test;
set test;
array nm(*) _numeric_;
do _n_ = 1 to dim(nm);
   nm(_n_) + 0;
end;
run;

# THERE IS A SHORTER WAY

proc stdize data=test out=new reponly missing=0;
run;

# CAN USE SAME NAME FOR OUT= ... WITH ALL VALUES
MISSING YOU GET A WARNING IN THE LOG THAT HAS
NOTHING  TO WITH THE VALUE REPLACEMENT

# WHAT VALUES WERE REPLACED

data test;
input name :  $10.  gender  :  $1.  age  height weight;
datalines;
MIKE . 21 60 .L
MARY .  0  . .H
MARK . 45 10  0
;

proc format;
value fix low‐high='0' other='1';
run;

filename nosee dummy;

data test;
file nosee;
set test;
put (_numeric_) (fix.) @;
missing = _file_;
put;
run;

proc stdize data=test out=new reponly missing=0;
run;

